
Kapitel 24

Reihenentwicklungen

Bei vielen Berechnungen treten Funktionen auf, die mathematisch nur mit sehr hohem
Aufwand (oder überhaupt nicht) weiterbearbeitet werden können. Vor allem in techni-
schen Anwendungen können solche Funktionen oft durch besser handhabbare Reihen-
entwicklungen ersetzt werden.

series

führt eine Taylor-, Laurent- oder eine verallgemeinerte Potenzreihenentwicklung durch.
Eine mögliche Anwendung von series besteht in der Lösung von Differentialgleichungen.

mtaylor und poisson

führen multivariable Taylor-Reihenentwicklungen durch.

powseries

bezeichnet ein Package, das verschiedene Kommandos zur Bearbeitung von formalen Po-
tenzreihen enthält.

numapprox

stellt ein Package dar, das Kommandos zur numerischen Berechnung von (zumeist ratio-
nalen) Näherungsfunktionen enthält.

Verweis: Bereits in Kapitel 15 wurde das Kommando eulermac behandelt, das eine Euler-
Maclaurin-Reihe für eine allgemeine Summenformel erstellt. Informationen zur Fourier-
Reihenentwicklung für periodische Funktionen finden Sie im nächsten Kapitel.

Taylor-, Laurent- und allgemeine Potenzreihenentwicklung 368

Taylor-, Laurent- und allgemeine Potenzreihenentwicklung

Das Standardkommando zum Aufstellen von Reihenentwicklungen lautet series. Es pro-
duziert je nach der Form der zu entwickelnden Funktion eine Taylor-, eine Laurent- oder
eine allgemeine Potenzreihe (jeweils auch komplex). Dem Kommando wird im ersten Pa-
rameter die zu entwickelnde Funktion und im zweiten Parameter der Entwicklungspunkt
übergeben. Im dritten Parameter kann die maximale Ordnung der Entwicklung angege-
ben werden. Wird darauf verzichtet, greift Maple dafür auf die globale Variable Order

zurück (Voreinstellung 6). Für die Entwicklung von Funktionen in Reihen und vor allem
für das Weiterarbeiten mit diesen Reihen ist es wichtig, sich mit den verschiedenen Typen
vertraut zu machen.

typen:=x->[type(x,taylor),type(x,laurent),type(x,series),whattype(x)];

series(sin(x),x,10); typen(%);

series(z/((z-(1+I))^2*(z-2)),z=1+I,2); typen(%);

series(sin(sqrt(x)),x,4); typen(%);

Weitere Beispiele finden Sie im Worksheet. Es gibt auch den Entwicklungspunkt 1 (in
diesem Fall ruft series das Kommando asympt auf, um die Reihenentwicklung durch-
zuführen).

series(arctan(x), x=infinity);

Weiterverarbeitung von Reihen

Mit series erhält man normalerweise ein Ergebnis im Datentyp series. Dabei wird eine
besonders effiziente Speicherung der Reihe (in Form des Entwicklungsorts und der Größe

Weiterverarbeitung von Reihen 369

der Koeffizienten) verwendet. Aus diesem Grund kann mit den Ergebnissen von series

nicht unmittelbar weitergearbeitet werden.

f:=series(sin(x), x=Pi);

f := �x+ � +

whattype(f);

series

Mit dem Kommando convert/polynom kann eine Umwandlung in ein normales Polynom
ohne Ordnungsterm durchgeführt werden. Dieses Polynom kann in der Folge zum Rech-
nen, Vereinfachen, Differenzieren, Zeichnen etc. verwendet werden.

p:=convert(f, polynom);

p := �x+ � +
(x� �)

Die Abbildung zeigt die Reihenentwick-
lung von sin(x) um den Punkt x = � und
die Originalfunktion.

plot({p, sin(x)}, x=0..2*Pi);

Ein weiterer Sonderfall tritt ein, wenn series bei einer Reihenentwicklung zu einem ex-
akten Ergebnis kommt. Das kann beim Rechnen mit Reihen vorkommen oder einfach so
erzeugt werden:

f:=series(x^2, x=3);

f := 9 + 6 (x� 3) + (x� 3)
2

simplify(f);

9 + 6 (x� 3) + (x� 3)
2

Da das Ergebnis exakt ist, entfällt der Ordnungsterm. Das Ergebnis sieht deswegen aus
wie ein normales Polynom. Dennoch wird es intern durch den series-Datentyp dargestellt
und muss mit convert in ein Polynom verwandelt werden, bevor die Vereinfachung zur
Ausgangsfunktion x

2 gelingt.

http://www.kofler.cc



Weiterverarbeitung von Reihen 370

whattype(f);

series

convert(f,polynom);

�9 + 6x+ (x� 3)
2

simplify(%);

x
2

Innerhalb von series können Sie mit Reihen ähnlich wie mit anderen mathematischen For-
meln arbeiten: Erlaubt sind nicht nur Operationen mit skalaren Größen (z.B. die Multipli-
kation der Reihe mit einem Faktor), sondern auch die Verknüpfung von mehreren Reihen
(Addition, Multiplikation ...). In den folgenden Beispielen werden unter anderem die Rei-
hen zu sin(x) und cos(x) miteinander multipliziert. Zum selben Ergebnis käme auch das
Kommando series(sin(x)*cos(x), x=Pi/2).

s1:=series(sin(x),x=Pi/2);

s2:=series(cos(x), x=Pi/2);

series(1/s1, x=Pi/2); #Reihe zu 1/sin(x)

series(s1+x^2, x=Pi/2); #Reihe zu sin(x)+x^2�

series(s1*s2, x=Pi/2); #Reihe zu sin(x)*cos(x)

s1:=series(sin(x),x=0);

series(arcsin(s1), x=0); # Umkehrfunktion

simplify(convert(%, polynom));

Differentialgleichungen mit Reihenentwicklungen lösen 371

Differentialgleichungen mit Reihenentwicklungen lösen

Eine praktische Anwendung von series besteht in der Lösung von Differentialgleichun-
gen durch einen Reihenansatz. Wir wollen das mit Maple an einer etwas ungewöhnlichen
DGL versuchen.

de:=y(x)+diff(y(x),x)=sin(sqrt(x));

dsolve({de,y(0)=0},y(x));

de := y(x) +
@

@x
y(x) = sin(

p
x)

y(x) = e
(�x)

Z x

0

sin(
p
u) e

u
du

Es gibt keine Stammfunktion

integral:=int(sin(sqrt(x))*exp(x),x);

integral :=

Z
sin(

p
x) e

x
dx ;

und deshalb auch keine Reihe:

series(integral,x);

Error, (in series/int) unable to compute series

Nun gibt es mehrere Möglichkeiten:

1. Die rechte Seite (Inhomogenität) in eine Reihe entwickeln.

sr:=convert(series(sin(sqrt(x)),x,9),polynom);

sr :=
p

de:=y(x)+diff(y(x),x)=sr:

sol:=rhs(dsolve({de,y(0)=0},y(x)));

Multivariable Taylor-Reihenentwicklung 372

Die Fehlerfunktion erf stört vielleicht etwas, also können wir nochmals in eine Reihe ent-
wickeln (siehe Worksheet) und erhalten folgende Lösung:

plot([sr,sol,psol],x=0..60,-1..1);

Dabei ist sr die rechte Seite der DGL, die
Lösung sol (Fehlerfunktion ohne Entwick-
lung) folgt ihr verzögert und die Entwick-
lung der Fehlerfunktion (psol) verursacht
starke Oszillationen. Die Lösung mit der
nicht entwickelten Fehlerfunktion ist dage-
gen (je nach Rechengenauigkeit) auch in ei-
nem zehnfach größeren Bereich stabil. –1

Zwei weitere Möglichkeiten finden Sie im Worksheet: 2. Entwicklung des Integranden, der
in dem von Maple nicht ausgewerteten Integral vorkommt und 3. Aufstellen einer benut-
zerdefinierten Reihe für die linke Seite der DGL (und Koeffizientenvergleich mit match).
Für welche Methode man sich entscheidet, wird letzten Endes immer eine Frage der ge-
forderten Genauigkeit und des Rechenaufwands sein.

Multivariable Taylor-Reihenentwicklung

Das Kommando mtaylor führt Taylorentwicklungen für multivariable Funktionen durch
(während series nur für Funktionen einer Variablen ausgelegt ist). Der Aufruf des Kom-
mandos erfolgt prinzipiell wie bei series, im zweiten Parameter muss allerdings eine Lis-
te der Entwicklungspunkte der betroffenen Variablen angegeben werden. Im optionalen
dritten Parameter wird wiederum die Ordnung der Reihenentwicklung angegeben. Be-
achten Sie, dass bei der Defaulteinstellung 6 bereits sehr umfangreiche Ausdrücke mit bis
zu 21 Termen entstehen. Das Kommando liefert das Ergebnis als normales Polynom ohne
Ordnungsterme.

mtaylor(log(x+y), [x=1, y=Pi], 3);

Das Kommando poisson stellt eine Variante zu mtaylor dar. Der wesentliche Unterschied
besteht darin, dass bei trigonometrischen Funktionen die Koeffizienten nach der kanoni-
schen Fourier-Form angeordnet werden. poisson ist im Gegensatz zu mtaylor nicht in der
Lage, Reihenentwicklungen um einen beliebigen Punkt durchzuführen. Außerdem ist die
Auswahl der Funktionen, für die überhaupt eine Entwicklung durchgeführt werden kann,
im Vergleich mtaylor stark eingeschränkt.

Formale Reihen 373

Das folgende Beispiel führt eine Reihenentwicklung für die Funktion sin(a+x)�cos(b+y)

sowohl mit poisson als auch mit mtaylor durch. Mit combine/trig lässt sich zeigen, dass die
beiden Ergebnisse identisch sind.

f := sin(a+x)*cos(b+y);

f1:=poisson(f, [x,y], 3 );

f1 :=

f2:=mtaylor(f, [x,y], 3);

f2 := sin(a) cos(b)� sin(a) sin(b)y + cos(a)x cos(b)�
sin(a) cos(b)y2

2
� cos(a)x sin(b)y�

sin(a)x2 cos(b)

2

combine(f1-f2,trig);

0

Formale Reihen

Das Package powseries enthält Kommandos zur Bearbeitung formaler Reihen. Damit sind
Reihen gemeint, die durch eine allgemeine Entwicklungsformel definiert werden. Das
Package powseries ermöglicht es, mit solchen Reihen zu rechnen. Erlaubte Rechenschrit-
te sind unter anderem Addition, Subtraktion, Multiplikation und Division zweier Reihen,
das Logarithmieren von Reihen, das Bilden von Ableitung und Integral, das Bilden der
inversen Reihe. Der Vorteil von formalen Reihen besteht darin, dass sie nach Ende der
Berechnung in beliebiger (!) Ordnung ausgewertet werden können.

with(powseries);

[compose ; evalpow ; inverse; multconst ; multiply ; negative ; powadd ; powcos ; powcreate ;

powdi� ; powexp; powint ; powlog ; powpoly ; powsin ; powsolve ; powsqrt ; quotient ;

reversion ; subtract ; template ; tpsform ]

Die beiden elementarsten Kommandos des Package lauten powcreate und tpsform. Mit
powcreate wird eine Prozedur erzeugt, die die berechneten Koeffizienten und das Bil-
dungsgesetz in der remember table ablegt.

Formale Reihen 374

powcreate( p(n)=1/n!);

seq(p(i),i=0..6);

1; 1;
1

2
;
1

6
;

1

24
;

1

120
;

1

720

p(100);

1=9332621544394415268169923885626670049071596826438162146859296389n
52175999932299156089414639761565182862536979208272237582511n
85210916864000000000000000000000000

Hier steht das Bildungsgesetz:

p(_k);

1

k !

Und die remember table kann so eingesehen werden:

op(4,op(p));

table([0 = 1; 1 = 1; 2 =
1

2
; 3 =

1

6
; 4 =

1

24
; 5 =

1

120
; 6 =

1

720
; k =

1

k !
;

100 = 1=93326215443944152681699238856266700490715968264381621n
46859296389521759999322991560894146397615651828625369792082n
7223758251185210916864000000000000000000000000])

So kann man sich auch die ganze Prozedur ausgeben lassen (Ausgabe hier nicht gedruckt):

interface(verboseproc=3): print(p); interface(verboseproc=1):

Die so gebildeten Koeffizienten können in der ’truncated powerseries form’ dargestellt
werden:

tpsform(p, x, 5);

Bevor eine neue Reihe mit dem alten Namen gebildet werden soll, sollte man den Namen
wieder freigeben. Die Koeffizienten können auch rekursiv und mit Anfangsbedingungen
angegeben werden:

p:='p': powcreate( p(n)=p(n-1)*p(n-2), p(0)=1, p(1)=1/2);

tpsform(p, x, 5);

Zur Übung können wir nun unsere eigene Taylorreihe aufstellen. Wenn man nicht um
x = 0 entwickelt, benötigt man allerdings einen kleinen Trick, um dies tpsform mitzuteilen.

powcreate(mytayl(n)=diff(sin(a),a$n)/n!,mytayl(0)=0):

seq(eval(mytayl(i),a=0),i=1..10);

1; 0;

6

tpsform(mytayl,x);

eval(tpsform(mytayl,x),x=x-a);

Wenn man die Taylorformel nicht verwendet, muss man unter Umständen dafür sorgen,
dass Koeffizienten Null werden:

powcreate(p(k)=1/k!*signum(sin(Pi/2*k)));

tpsform(p,x,9); #series(sin(x),x);

powcreate(q(k)=1/k!*signum(cos(Pi/2*k)));

Nun kann mit den Reihen gerechnet werden. Für algebraische Operationen verwendet
man evalpow, für die Umkehrung reversion, für die Ableitung powdiff usw., wobei diese
Befehle jeweils wieder eine kleine Prozedur erzeugen:

r:=evalpow(p*q);

r := proc(powparm) : : : end proc

tpsform(r,x,9);

series( sin(x)*cos(x), x, 9);

r:=reversion(p);

series( arcsin(x), x, 9);

r:=powdiff(p); tpsform(r,x,9);

Die Koeffizienten kann man sich auch anders besorgen: Die Übernahme von Reihen, die
mit series gebildet wurden, erfolgt mit powpoly. Dabei kann man leicht ’beliebig viele’
Koeffizienten zur Verfügung stellen:

s:=series( tan(x), x=0, 100):

r:=powpoly( convert(s,polynom), x ):

t:=evalpow(p+r): t(81);

41752740333315135109856335029366067957732854320613176020120487602n
606216432104998754655524723365930416039

Æ
252048957423798n

60808172757528600474371358074977094025907590717607462565998n
6690958967660197141293301760000000000000000000

tpsform(t,x,15);

series(tan(x)+sin(x),x,15);

Numerische Berechnung von Näherungsfunktionen

Das Package numapprox fasst verschiedene Funktionen zur Berechnung von Näherungs-
funktionen zusammen:

with(numapprox); with(orthopoly):

[chebdeg ; chebmult ; chebpade ; chebsort ; chebyshev ; confracform ; hermite pade ;

hornerform ; infnorm ; laurent ; minimax ; pade ; remez ]

� minimax erstellt eine rationale Näherungsfunktion, die die gegebene Funktion in einem
Bereich (x0::x1) annähert.

Numerische Berechnung von Näherungsfunktionen 377

� pade ist die Pade-Approximation zur Berechnung einer rationalen Näherungsfunktion.
Dabei wird wie bei series um einen Punkt x0 entwickelt.

� chebpade berechnet ebenfalls eine rationale Näherungsfunktion für den Bereich
(x0::x1). Die Näherungsfunktion ist aus Tschebyscheff-Polynomen zusammengesetzt.

� chebyshev berechnet eine Tschebyscheffsche Reihe für die angegebene Funktion im Be-
reich x0::x1.

Daneben enthält das Package einige weitere Kommandos: taylor und laurent stellen ein-
geschränkte Varianten des series-Kommandos dar. hornerform und confracform führen
Polynome bzw. rationale Funktionen in die Hornerform bzw. in einen Kettenbruch über –
siehe convert/horner bzw. /confrac in Kapitel 10. remez wird von minimax zur Optimierung
der rationalen Funktion verwendet und wird nur in Sonderfällen direkt vom Anwender
aufgerufen.

Anmerkungen: Die Rechenzeit der Kommandos ist fallweise ziemlich hoch. Eine Reduzie-
rung von Digits kann erhebliche Verbesserungen bringen. Wenn die Kommandos statt
eines Ergebnisses eine Fehlermeldung liefern, können Sie versuchen, eine kleine Verände-
rung der Ausgangsparameter (inklusive der Rechengenauigkeit) vorzunehmen.

In den folgenden Beispielen werden die verschiedenen Algorithmen verglichen. Als
’Härtetest’ können wir z.B. die Funktion tan(x2) mit der Unstetigkeit (Pol) im Punkt
x0 =

p
�=2 nehmen. Wie üblich können Sie im Worksheet andere Funktionen einsetzen

und alle Parameter ändern.

Wir definieren die Funktion, die Unstetig-
keitsstelle und Bereichsangaben.

f:=x->tan(x^2);

f := x! tan(x
2
)

x0:=sqrt(Pi/2): a:=0: b:=1:

Nun kann mit series eine Laurentreihe um
x0 und eine Taylorreihe um den Ursprung
entwickelt werden.

sx0:=series(f(x),x=x0,40):

s:=series(f(x),x=a,40):

plot([f(x),convert(s,polynom),convert(sx0,polynom)],x=-2..3,-10..10,);

Die Laurentreihe ist am Pol nicht von
der Funktion zu unterscheiden, hat aber
für kleine x eine sehr große Abweichung
von f(x). Die Taylorreihe approximiert die
Funktion für kleine x sehr gut, gibt aber jen-
seits der Pole die Funktion nicht wieder.

Nun können wir untersuchen, inwiefern die Befehle aus dem Paket numapprox eine Verbes-
serung der Annäherung bringen. Die Prozedur minimax approximiert nach dem Remes-
Algorithmus. Die erste Variante mini0 ist ein Polynom fünften Grades. Bei der zweiten
Variante mini25 hat der Zähler den Grad 2 und der Nenner den Grad 5.

mini0:=minimax(f(x),x=a..x0-0.1,5,1,'fehler'); expand(mini0); fehler;

mini0 := �:100575884 + (4:867599482

+ (�36:71069950 + (102:4890774 + (�114:2815642 + 45:21405146 x)x)x)x)x

�:100575884 + 4:867599482 x� 36:71069950 x
2
+ 102:4890774 x

3 � 114:2815642 x
4

+ 45:21405146 x
5

fehler = :1006673773

Mit der gebrochen rationalen Funktion (Ausgabe hier unterdrückt) erhält man eine we-
sentlich bessere Näherung:

mini25:=minimax(f(x),x=a..x0-0.05,[2,5],1,'maxerror'); maxerror;

maxerror = :0005507722888

plot([f(x),mini0,mini25],x=-1/2..3,-10..10);

Die Annäherung ist bei den hier verwende-
ten Parametern schlechter als mit der nor-
malen Reihenentwicklung. Wenn man mit
der rechten Bereichsgrenze zu nahe an x0

geht, beginnt das Näherungspolynom zu
oszillieren.

Als Nächstes können wir die beiden Sorten der Pade-Approximation untersuchen.
Zunächst eine Enwicklung um den Ursprung:

padef:=pade(f(x),x=a,[6,17]);

padef :=
�

10

99
x
6 + x

2

1�
43

99
x4 +

17

1485
x8 +

4

31185
x12 +

1

467775
x16

Und dann eine Tschebyscheff-Pade-Approximation von Pol zu Pol (Ausgabe im Work-
sheet):

chebpadef1:=chebpade(f(x),x=-x0+0.1..x0-0.1,[6,17]):

plot([f(x),padef,chebpadef1],x=-3..3,-10..10);

Die Pade-Approximation liefert hier mit
dem geringsten Aufwand das beste Er-
gebnis, indem sie sowohl die Symmetrie
als auch noch den zweiten Pol wieder-
gibt. Insofern ist also hier der Einsatz von
Tschebyscheff-Polynomen überflüssig.

Schließlich folgt noch eine Demonstration, wie man es nicht machen sollte:

chebpadef2:=chebpade(f(x),x=a..1.25,[7,0]);

cheby:=chebyshev(f(x),x=a..1,0.001);

plot([f(x),chebpadef2,cheby],

x=-1/2..2,-10..10);

Die mit chebpade gebildete Reihe beginnt
sehr stark zu oszillieren, wenn man sich mit
der Bereichsgrenze dem Pol nähert. Die mit
chebyshev gebildete Reihe ist dagegen zwar
stabil, aber eben nur im Intervall von 0 bis
1 eine gute Näherung.

Syntaxzusammenfassung

series(f,x); series(f,x=x0,n);

führt eine Laurent-, Taylor- oder allgemeine Potenzreihenentwicklung für f um x = 0

oder um x = x0 bis zur Ordnung Order oder bis zur im dritten Parameter angegebe-
nen Ordnung n durch.

asympt(f,x); asympt(f,x,n);

führt eine Reihenentwicklung für x!1 bis zur Ordnung n durch.

whattype(%);

stellt den Datentyp des letzten Ergebnisses fest. Nach den Kommandos series und
asympt kommen series oder�+�in Frage.

convert(reihe, polynom);

konvertiert eine Reihe vom Datentyp series in ein Polynom.

Order:=n;

bestimmt die Defaulteinstellung für die Ordnung, bis zu der Reihen entwickelt wer-
den sollen.

mtaylor(f,[x=x0,y=y0,...],n);

führt eine multivariable Taylor-Reihenentwicklung für f = f(x; y; :::) durch.

poisson(f,[x=x0,y=y0,...],n);

wie oben, allerdings werden die Koeffizienten in der kanonischen Fourier-Form an-
geordnet.

Formale Reihen

powcreate(f(n)=...);

definiert die formale Reihe f durch eine Funktion für die Koeffizienten zu x
0
; x; x

2
;

x
3
; :::. Die Koeffizienten können mit f(0), f(1), ... abgerufen werden.

tpsform(f, x, n);

gibt die formale Reihe f als Reihe (Datentyp series) der Variable x bis zur Ordnung
n zurück.

powpoly(polynom, x);

verwandelt ein Polynom in eine formale Reihe.

evalpow(f +-*/ g); evalpow(f */^ s); evalpow(f(g));

führt elementare Berechnungen mit den Reihen f und g und dem Skalar s durch (Ad-
dition, Subtraktion, Multiplikation, Division etc.).

powdiff(f); powint(f);

differenziert bzw. integriert die Reihe.

powlog(f); powexp(f);

bildet den Logarithmus und die Exponentialfunktion der Reihe.

reversion(f);

bildet die Reihe zur Umkehrfunktion.

Numerische Berechnung von Näherungsfunktionen

minimax(f, x=a..b, [z,n]);

stellt eine rationale Funktion mit dem Zählergrad z und dem Nennergrad n auf, die
die Funktion f im Bereich zwischen a und b annähert.

pade(f, x=x0, [z,n]);

stellt eine rationale Näherungsfunktion auf, die f im Bereich um x = x0 annähert.

chebpade(f, x=a..b, [z,n]);

wie oben, allerdings wird die Funktion aus Tschebyscheff-Polynomen T (n; x � x0 )

zusammengesetzt.

chebyshev( f, x=a..b, eps);

berechnet eine Reihenentwicklung aus Tschebyscheff-Polynomen, die f im Bereich
zwischen a und b mit einer maximalen Abweichung von eps annähert.

T(n,x);

gibt das Tschebyscheff-Polynom n-ter Ordnung zurück. Zum Arbeiten mit diesen
Polynomen kann das Paket orthopoly aufgerufen werden. Achtung: Dadurch werden
Variablen mit den Namen G;H;L; P; T; U überschrieben.

